

6

Advanced Engineering Computations

Low-Pass Electrical Filter

Electrical engineers use filters extensively in radios, telephones, televisions, computers, power supplies, and many other practical devices. A specific example is the low-pass filter used in the design of a direct current (DC) power supply. An *ideal low-pass filter* is a device that completely transmits a signal unattenuated at all frequencies below a certain cutoff frequency and completely blocks those frequency components of the input signal above this cutoff frequency. The application in this chapter uses *complex phasors* (a common method used to describe a time-periodic quantity in terms of a single complex number) and *nodal analysis* (a circuit analysis technique described later in this chapter) to analyze an approximation to an ideal low-pass filter. This technique is very general, and is a powerful tool for circuit analysis. Maple's ability to find complex-valued solutions to algebraic equations forms the foundation upon which the remainder of the analysis can be completed. You will also learn Maple commands and techniques for use with more sophisticated mathematics concepts, including calculus and differential equations.

INTRODUCTION

Maple can be used in many different branches of mathematics including algebra, calculus, combinatorics, financial mathematics, graph theory, linear algebra, logic, number theory, and optimization. The first five chapters of this module showed you how to use Maple to solve problems requiring mainly algebra and trigonometry. In this chapter, you will see some of the ways Maple can be used to solve problems that involve complex analysis, calculus, and (briefly) differential equations. The application in this chapter shows how an electrical engineer uses nodal analysis to investigate some of the characteristics of a low-pass electrical filter. Recall that the bandwidth application in Chapter 3 also involved the analysis of a filter. Even though there are similarities in how engineers discuss both types of filters, the discussion in this chapter is quite different from the one presented in Chapter 3.

Although you will learn many useful techniques for solving mathematical problems in this chapter, it is not comprehensive in its presentation of Maple's mathematical capabilities. Some of the examples and Try It! exercises refer to problems encountered earlier in the module. For example, the solution to the Streeter-Phelps equation that was provided in Chapter 5 will now be found and verified using Maple. References to the earlier discussions are provided; if you do not recall the details of a problem, please take a few minutes to refresh your memory.

Several dozen packages contain collections of additional Maple commands. The **plots** and **student** packages have been introduced previously. The **student** package will be used in the discussion of calculus (Section 6-2) and the **DEtools** package will be used to produce plots relating to differential equations (Section 6-3). Other packages likely to be of interest to many engineers include the **linalg** package for linear algebra, the **numtheory** package for number theory, and the **inttrans** package for integral transformation such as the Laplace and Fourier transform. The full list of Maple packages can be found on the online help worksheet with keyword **index, package**.

6-1 COMPLEX NUMBERS

Maple functions and variables are, by default, assumed to be complex valued. This may seem to be an unnecessary feature for most standard computations, but it is actually one of the underlying characteristics that gives Maple its power and broad applicability. There are times, however, when this increased functionality interferes with some seemingly simple operations.

Basic arithmetic operations (addition, subtraction, multiplication, and division) are performed exactly as they would be for real-valued quantities. The only difference is the use of **I** to represent the square root of -1 , that is, $\mathbf{I} = \sqrt{-1}$. For example, the sum and product of $z = 2 + 3i$ and $w = 6 - 4i$ can be computed using

```
> z := 2+3*I;
```

$$z := 2 + 3 I$$

> **w := 6-4*I;**

$$w := 6 - 4I$$

> **SUMzw := z+w;**

$$SUMzw := 8 - I$$

> **QUOTzw := z/w;**

$$QUOTzw := \frac{1}{2}I$$

Common commands specifically designed for use with complex-valued objects include **abs**, **conjugate**, **Re**, and **Im**. For expressions involving symbolic names, the **evalc** command is used to force the evaluation of complex-valued expressions in the standard form: $x + Iy$.

EXAMPLE 6-1

Arithmetic of Complex Numbers

Let z denote a complex number with real part Rez and imaginary part Imz . Similarly, let $w = Rew + i Imw$. Find, in standard form, the *conjugate* of z , \bar{z} , and the real and imaginary parts of the product zw .

SOLUTION

The two complex numbers are

> **z := Rez + I*Imz;**

$$z := Rez + I Imz$$

> **w := Rew + I*Imw;**

$$w := Rew + I Imw$$

The complex conjugate of z , \bar{z} , simply changes the sign of the imaginary part:

> **zC := conjugate(z);**

$$zC := \overline{Rez + I Imz}$$

> **zC := evalc(zC);**

$$zC := Rez - I Imz$$

Note how **evalc** is used to reduce the expression into standard form.

The product of z and w is

> **PRODzw := z * w;**

$$PRODzw := (Rez + I Imz)(Rew + I Imw)$$

The real and imaginary parts of the product are

> **REzw := Re(PRODzw);**

$$REzw := \Re((Rez + I Imz) (Rew + I Imw))$$

> **IMzw := Im(PRODzw);**

$$IMzw := \Im((Rez + I Imz) (Rew + I Imw))$$

Or, in standard form,

> **REzw = evalc(REzw);**

$$\Re((Rez + I Imz) (Rew + I Imw)) = Rez Rew - Imz Imw$$

> **IMzw = evalc(IMzw);**

$$\Im((Rez + I Imz) (Rew + I Imw)) = Rez Imw + Imz Rew$$

Of course, the same information can be obtained by directly converting the product into standard form:

> **PRODzw = evalc(PRODzw);**

$$(Rez + I Imz) (Rew + I Imw) = Rez Rew - Imz Imw + I (Rez Imw + Imz Rew)$$

.....

Other operations work in a similar manner. One point that should be emphasized is that it is important to use built-in Maple commands for standard operations whenever possible. This is the point of the next Try It! exercise.

Try It



The *modulus*, or *magnitude*, of a complex number is the square root of the product of the number and its conjugate. Compute the modulus of the complex numbers z and $z w$ both from the definition and using Maple's **abs** command. Explain and resolve any differences in the appearance of the results.

You may have noticed that expressions returned by **evalc** (in standard form) are valid only when the real and imaginary parts of z and w (that is, Rez , Imz , Rew , Imw) are real-valued. In fact, the fundamental assumption made by **evalc** is that unassigned variables represent real-valued quantities. For example, when neither **a** nor **b** is assigned a value, **evalc(Re(a+I*b))** evaluates to **a**. Furthermore, **evalc** assumes that an unknown function of a real variable is real-valued. For additional explanation and examples, see the online help document for **evalc**.

Complex-valued quantities arise in any number of situations. One of the more common situations is when finding the roots of a polynomial. Recall that the Fundamental Theorem of Algebra states that any polynomial of degree n has exactly n roots. In addition to the quadratic formula, there are explicit formulae for all roots of any polynomial whose degree does not exceed four. Even though it has been proved that there is no general formula in terms of radicals for the root of general polynomials of degree higher than four, there are numerous special cases where Maple is still able to find some or all of the roots.



Try It Use Maple to find the formulae for the roots of a general quadratic, cubic, and quartic polynomial. Verify that Maple returns the appropriate number of roots. Note that the **allvalues** command can be used to force evaluation of results that involve **RootOf**.

Many mathematical functions have a natural extension to complex-valued arguments. For example, the definition of the exponential of a complex number is based on Euler's formula:

$$e^{(r + i\theta)} = e^r(\cos(\theta) + i \sin(\theta))$$

(In fact, many of the special functions, for example, **exp**, **ln**, **sin**, **cos**, **tan**, actually arise more naturally in this setting.)

In most cases, the complex-valued extensions of the functions are of little concern. However, one place where this cannot be ignored is when working with fractional powers. Although most humans immediately simplify $(-1)^{1/3}$ to -1 , Maple realizes that there are three possible values to this expression. Maple does choose one of the three values, but—for mathematical reasons beyond the scope of this module—it is *not* the real-valued root. The **surd** command provides a means of obtaining a real-valued output from odd roots of negative numbers:

```
> (-8)^(1/3) = evalc( (-8)^(1/3) );
```

$$(-8)^{1/3} = \frac{1}{2}8^{1/3} + \frac{1}{2}I8^{1/3}\sqrt{3}$$

```
> (-8)^(1/3) = surd( -8, 3 );
```

$$(-8)^{1/3} = -2$$

EXAMPLE 6-2

Plotting Fractional Powers

Plot the function $y = x^\alpha$ for $\alpha = \frac{1}{2}$, $\frac{1}{3}$, $\frac{2}{3}$, $\frac{1}{4}$, $\frac{2}{4}$, and $\frac{3}{4}$ on the interval $[-1, 1]$.

SOLUTION


The preceding discussion suggests that the naive approach will not be successful. The following plot confirms that each of these functions is, in this form, real-valued only when $x \geq 0$.

```
> FNS1 := [ x^(1/2), x^(1/3), x^(2/3), x^(1/4), x^(2/4), x^(3/4) ];
```

$$FNS1 := [\sqrt{x}, x^{1/3}, x^{2/3}, x^{1/4}, \sqrt{x}, x^{3/4}]$$

```
> plot( FNS1, x=-1..1, color=[RED, RED, BLUE, RED, BLUE, GREEN],
>      symbol=CIRCLE, style=[POINT, LINE$5],
>      title=`Six Fractional Powers` );
```

Six Fractional Powers



Note that each plot starts at $x=0$, not $x=-1$, as requested. This is not a surprise for the even roots since, for example, $(-1)^{1/2} = \sqrt{-1} = i$, which is not real valued. Note also that the plots with $\alpha = 1/2$ and $\alpha = 2/4$ are identical.

To obtain the desired plot, each of the functions must be represented in terms of **surd**:


```
> FNS2 := [ surd(x,2), surd(x,3), surd(x^2,3), surd(x,4),
>          surd(x^2,4), surd(x^3,4) ];
```

$$FNS2 := [\text{surd}(x,2), \text{surd}(x,3), \text{surd}(x^2,3), \text{surd}(x,4), \text{surd}(x^2,4), \text{surd}(x^3,4)]$$

Note the steps taken to comply with the requirement that the second argument to **surd** be an integer (see the online help for **surd**).

```
> plot(FNS2, x=-1..1, color=[RED,RED,BLUE,RED,BLUE,GREEN],
>      symbol=CIRCLE, style=[POINT,LINE$5],
>      title=`Six Fractional Powers -- using surd` );
```

Six Fractional Powers -- using surd



From the second plot, you can see that using **surd** allowed three of the plots to be extended to the negative real line.

.....



Identify the three functions that were able to be extended to the negative real line in the final plot in Example 6-2.



Use **convert** and **simplify** (and, if necessary, **assume**) to verify that the **surd**-based representations are equivalent to the original power representation. (Pay particular attention to $\alpha = \frac{2}{4}$.)

To conclude this brief introduction to complex analysis, it should be mentioned that the **plots** package contains commands for creating 2D and 3D plots of complex-valued functions: **complexplot** and **complexplot3d**.

EXAMPLE 6-3**Complex-Valued Solutions of an Equation**

Find, and plot, the ninth roots of -10 .

SOLUTION

The ninth roots of -10 are the nine solutions to the ninth-degree polynomial

```
> EQ := x^9 = -10;
```

$$EQ := x^9 = -10$$

```
> ROOTS := [ solve( EQ, x ) ];
```

$$ROOTS := \left[-10^{1/9}, \frac{1}{2}10^{1/9} - \frac{1}{2}I\sqrt{3} 10^{1/9}, \frac{1}{2}10^{1/9} + \frac{1}{2}I\sqrt{3} 10^{1/9}, \frac{1}{2}\%2^{1/3}, -\frac{1}{4}\%2^{1/3} - \frac{1}{4}I\sqrt{3} \%2^{1/3}, \right. \\ \left. -\frac{1}{4}\%2^{1/3} + \frac{1}{4}I\sqrt{3} \%2^{1/3}, \frac{1}{2}\%1^{1/3}, -\frac{1}{4}\%1^{1/3} + \frac{1}{4}I\sqrt{3} \%1^{1/3}, -\frac{1}{4}\%1^{1/3} - \frac{1}{4}I\sqrt{3} \%1^{1/3} \right]$$


$$\%1 := 4 10^{1/3} + 4I\sqrt{3} 10^{1/3}$$

$$\%2 := 4 10^{1/3} - 4I\sqrt{3} 10^{1/3}$$

It can be difficult to see any structure in the preceding representations of the roots. A point plot of these numbers, in the complex plane, should yield more information:

```
> plots[complexplot]( ROOTS, style=POINT, labels=['Re', 'Im'],
> axes=FRAMED, title=`Ninth roots of -10` );
```

Ninth roots of -10



Note that there is only one real-valued root, and none of the roots are purely complex. Moreover, the roots appear to be regularly spaced on the circle centered at the origin with radius a little larger than 1.25.

.....

Application 6

NODAL ANALYSIS OF AN ELECTRICAL CIRCUIT

Electrical engineers have, as one of their tasks, the analysis and synthesis of a variety of circuits. In this application, you will use Maple to conduct an analysis of a low-pass (LP) electrical filter. An ideal LP filter transmits an input electrical signal to the output with no attenuation and no phase shift. However, above some *cutoff frequency*, ω_c , the electrical signal is completely blocked and usually undergoes some phase shift as well. The cutoff frequency can be controlled by the proper choice of component values.


No practical filter behaves in such an ideal fashion, but this situation can be approximated as closely as cost and need dictate. Your analysis of a passive low-pass filter consisting of two inductors and one capacitor will show that this filter exhibits resonant (sharp peak) behavior in the vicinity of the cut-off frequency, which detracts from the ideal uniform behavior. Nevertheless, its rejection of frequencies beyond the cutoff frequency is a factor of 100 higher than for a simpler (one inductor or one capacitor) circuit. Stated in terms of decibels, this improvement by a factor of 100 corresponds to $20 \log_{10}(100) = 40$ decibels (dB). In practice, an LP filter would likely be constructed from resistors (modeling loss of power) and active circuit elements such as operational amplifiers.

Fundamentals

The circuit diagram in Figure 6-1 shows a set of input terminals (the input port) and a set of output terminals (the output port). The dashed box encloses the portion of the circuit designated the low-pass filter. The input port has a voltage signal $v_i(t)$ across its terminals, and the output port similarly has output voltage $v_o(t)$. Attached to the output terminals is an electrical load with resistance, R . There is an input AC voltage source across the input terminals, with a 170-volt peak amplitude, 120° phase shift, and oscillating at the radian frequency of ω radians/second ($= 2\pi 60$ for AC voltages available from wall outlets in the United States). That is, the input source is $v_i(t) = 170 \cos(\omega t + 2/3\pi)$. (This represents a standard 120-volt rms AC voltage available from wall outlets.)

Time-periodic (or steady-state sinusoidal) sources are so prevalent in the analysis of electrical circuits (as well as other systems in engineering) that it is common to describe them in terms of the peak amplitude and phase shift (in degrees), that is, as a *phasor*. For example, the phasor representation of the input voltage is $170 \angle 120^\circ$, in units of Volts (note that $2\pi/3$ radians = 120°). Conversion between the phasor and time-

Figure 6-1
Low-pass filter




periodic representations is based on Euler's formula, $v_i(t) = \Re(V_i e^{j\omega t})$, where now $j = \sqrt{-1}$ (as is standard engineering practice).

The ratio of the phasor voltage to the phasor current associated with any element gives the impedance of that element. The impedance of a capacitor is $Z_c = -j/\omega C$, where C is the capacitance (in Farads) of the capacitor. The impedance of an inductor with inductance L (in Henries) is $Z_L = j\omega L$. So, as is shown in Figure 6-2, if the phasor voltage across the capacitor is V_C , then a phasor current V_C/Z_c flows through the capacitor. Similarly, the phasor current flowing through the load resistor is V_o/R .

A *node* is a location where two or more circuit elements are connected. In nodal analysis, each node in the circuit has an associated *node voltage*. In addition to the common (reference) node at the bottom of the circuit, the three nodes in the circuit shown in Figure 6-1 have voltages of V_i , V_C , and V_o . The node voltage V_i is known from the voltage source, so only the voltages V_C and V_o need to be determined. This is summarized in the *impedance-phasor circuit diagram* shown in Figure 6-2.

Figure 6-2
Impedance equivalent of a low-pass filter



Two equations are needed to determine the two unknown node voltages at nodes 1 and 2. These equations are obtained by applying *Kirchhoff's current law* at each node. Kirchhoff's current law is, basically, conservation of charge, that is, the sum of all currents leaving (or entering) any node must be zero. The node to which the capacitor is connected (node 1) has three paths of current: two through the two inductors and one through the capacitor. The other node (node 2) has two paths where current can flow: one through the load resistor and the other through the output inductor. This yields the two nodal equations. The unknown node voltages are found by simultaneously solving this pair of equations.

1. Define the problem

Develop an expression for the ratio of the output and input voltages, that is, the *transfer voltage ratio*, as a function of frequency ω (in radians per second) for the LP filter shown in Figure 6-2. The origin of the name low-pass filter can be seen by examining a plot of this ratio. Make a *corrected Bode amplitude plot*, that is, a plot of the magnitude of the transfer voltage ratio (in decibels) as a function of $\log(\omega/\omega_0)$ where $\omega_0^2 = 1/LC$. When numerical values are needed, and are otherwise unspecified, use $L = 10\text{H}$, $C = 6\ \mu\text{F}$, and $R = 1\ \text{k}\Omega$.

2. Gather information

Kirchhoff's current law (KCL) is the underlying principle on which this application is developed. The first step in the analysis is, therefore, to determine the two nodal equations in terms of general values of the inductance, L , capacitance, C , and resistance, R .

The phasor current leaving node 1 through the input inductor is the ratio of the voltage difference across its terminals, $V_C - V_i$, to the

impedance, $j\omega L$, of this element: $\frac{V_C - V_i}{j\omega L}$. Similarly, the phasor currents

leaving node 1 through the capacitor and output inductor are $\frac{V_C}{-j/\omega C}$

and $\frac{V_C - V_o}{j\omega L}$, respectively. Thus, applying KCL at the first node means

that the sum of all currents leaving node 1 is zero:

$$\frac{V_C - V_i}{j\omega L} + \frac{V_C}{-j/\omega C} + \frac{V_C - V_o}{j\omega L} = 0$$

The second node connects only two elements. The current phasor leaving the output inductor is the negative of the current entering that element: $\frac{V_o - V_C}{j\omega L}$; the current phasor through the resistor is $\frac{V_o}{R}$. The equation obtained from applying KCL at node 2 is

$$\frac{V_o - V_C}{j\omega L} + \frac{V_o}{R} = 0$$

3. Generate and evaluate potential solutions

In order to be faithful to engineering notation, it is preferable to override Maple's use of **I** to represent $\sqrt{-1}$. The online help for **I** tells you that **I** is simply an alias for $\sqrt{-1}$. Then, following the hyperlink to the help worksheet for **alias**, you quickly learn that the following command both removes the alias for **I** and adds a new alias for **j**.

```
> restart;
> alias( I=I, j = sqrt(-1) );
```

j

The equations representing KCL at the two nodes are

```
> EQ1 := (V[C]-V[i])/j/omega/L + V[C]/(-j/omega/C) +
> (V[C]-V[o])/j/omega/L = 0;
```

$$EQ1 := -\frac{j(V_C - V_i)}{\omega L} + jV_C\omega C - \frac{j(V_C - V_o)}{\omega L} = 0$$

```
> EQ2 := (V[o]-V[C])/(j*omega*L) + V[o]/R = 0;
```

$$EQ2 := -\frac{j(V_o - V_C)}{\omega L} + \frac{V_o}{R} = 0$$

The voltage transfer ratio as a function of frequency (or, simply, the frequency response) for this circuit is the ratio of the output and input voltage phasors: $\frac{V_o}{V_i}$. This expression is easily constructed once V_C and V_o are known.

```
> SOLN := solve( { EQ1, EQ2 }, { V[C], V[o] } );
```

$$SOLN := \left\{ V_C = -\frac{V_i(jR - \omega L)}{-jR + jR\omega^2 CL + 2\omega L - \omega^3 L^2 C}, V_o = -\frac{jRV_i}{-jR + jR\omega^2 CL + 2\omega L - \omega^3 L^2 C} \right\}$$

Although these solutions are complicated functions of frequency, it is not difficult to see that these voltages respond linearly to a change in the input voltage. The ratio of the output and input voltages is, therefore, independent of the input voltage, V_i :

```
> RESPONSE := subs( SOLN, V[o]/V[i] );
```

$$RESPONSE := \frac{jR}{-jR + jR\omega^2CL + 2\omega L - \omega^3L^2C}$$

Note that the frequency response is, for most frequencies, a complex number. If it is written in polar form, $\frac{V_o}{V_i} = M e^{j\theta}$, the magnitude M is commonly called the *gain*, and θ is called the *relative phase shift* (relative to the phase of the input source). The gain is found to be

```
> M := evalc( abs( RESPONSE ) );
```

$$M := \sqrt{\frac{R^2(2\omega L - \omega^3L^2C)^2}{\left((2\omega L - \omega^3L^2C)^2 + (-R + R\omega^2CL)^2\right)} + \frac{R^2(-R + R\omega^2CL)^2}{\left((2\omega L - \omega^3L^2C)^2 + (-R + R\omega^2CL)^2\right)}}$$

which can be simplified, with **normal** (because of the rational functions that appear in M) and **collect** (to show the structure relative to ω) to the somewhat more manageable form

```
> M := map( collect, normal( M ), omega );
```

$$M := \sqrt{\frac{R^2}{\omega^6L^4C^2 + (-4L^3C + R^2C^2L^2)\omega^4 + (4L^2 - 2R^2CL)\omega^2 + R^2}}$$

(The **map** command is needed to force the collection with respect to ω in the numerator and denominator of this expression. Compare this result with that obtained from **collect(normal(M), omega);**.)

The phase angle is the argument of a complex-valued frequency response function. Maple's **argument** is used in the same manner as **abs**, **Re**, and **Im**:

```
> theta := argument( RESPONSE );
```

$$\theta := \text{argument}\left(\frac{jR}{-jR + jR\omega^2CL + 2\omega L - \omega^3L^2C}\right)$$

The argument is, essentially, the arctangent of the ratio of the imaginary and real parts of the expression. (Exceptions occur when the real part is zero; furthermore, the signs of the real and imaginary parts must be used to determine the correct quadrant for the argument; see the online help for **argument** and **invtrig** for further details.)

4. Refine and Implement a Solution

A useful tool in analyzing an electrical circuit is a plot of the corrected Bode amplitude. That is, a plot of the amplitude gain, in dB,

$M_{dB} = 20 \log_{10}(M)$, versus $\log_{10}(\omega/\omega_0)$ where the natural (radian) frequency,

ω_0 , satisfies $\omega_0^2 = 1/LC$. Thus,

```
> omega0 := sqrt(1/(L*C));
```

$$\omega_0 := \sqrt{\frac{1}{LC}}$$

```
> MdB := 20*log[10](M);
```

$$MdB := 20 \frac{\ln \left(\frac{R^2}{\sqrt{\omega^6 L^4 C^2 + (-4L^3 C + R^2 C^2 L^2) \omega^4 + (4L^2 - 2R^2 CL) \omega^2 + R^2}} \right)}{\ln(10)}$$

The formula for the amplitude gain can be simplified using the properties of logarithms. Before Maple will apply these properties, it is necessary to tell Maple that all the variables are real valued:

```
> assume(R>0, L>0, C>0, omega>0);
```

```
> MdB := simplify(MdB);
```

$$MdB := \frac{2 \ln(R) - \ln(\omega^6 L^4 C^2 - 4 \omega^4 L^3 C + \omega^4 R^2 C^2 L^2 + 4 \omega^2 L^2 - 2 \omega^2 R^2 C L + R^2)}{\ln(2) + \ln(5)}$$

The corrected Bode amplitude plot can be constructed using the **semilogplot** command from the **plots** package. Since the horizontal axis should be normalized to the natural frequency, it is most convenient to specify the curve in parametric form. The specific form that is most appropriate for use with **semilogplot** is

```
> BODE := [ omega/omega0, MdB, omega=omega0/10 .. omega0*10 ]:
```

According to the help worksheet for **semilogplot**, this should plot $\log_{10}(\omega/\omega_0)$ against M_{dB} as desired.


The plot will be created using the default set of parameter values for the components of the filter:

```
> PARAM := [ L = 10, C = 6*10^(-6), R = 10^3 ];
```

$$PARAM := \left[L = 10, C = \frac{3}{500000}, R = 1000 \right]$$

The corrected Bode amplitude plot corresponding to these parameter values is

```
> with(plots):
> BODEp := semilogplot( subs( PARAM, BODE ), axes=FRAMED,
>                       labels=[`omega/omega0`, `gain (dB)` ],
>                       title=`corrected Bode amplitude plot` ):
> display( BODEp );
```



Observe that this curve is essentially linear when $\omega > 2\omega_0$. Since the slope is approximately -60 , this circuit is said to exhibit a rolloff of 60 dB per decade (factor of 10 in frequency) or 18 dB per octave (factor of 2 in frequency).

This corrected Bode amplitude plot can be easily approximated by zero dB for low frequencies and by a linear function (in $\log_{10}(\omega/\omega_0)$) for high frequencies. The corresponding *uncorrected* Bode amplitude plot consists of these two straight lines. The frequency at which these lines intersect is called the *cutoff frequency* (corner frequency and break frequency are other names that appear in the literature). The cutoff frequency for this low-pass filter is $\omega_c = 119$ rad/sec, which corresponds to $f_c = 19$ Hz. So you can see that this low-pass filter could be used, for example, as a component of a DC power supply, since it would reject any surviving AC signals at $f = 60$ Hz from household outlets and transmit only the desired DC ($f = \omega = 0$) voltage signals that the power supply generates.

5. Verify and Test the Solution

The significance of the natural frequency, ω_0 , is uncovered by examining the gain and phase shift. Other special cases of interest are $\omega = 0$ (DC signal), the 3 dB points (frequencies at which the magnitude of the output voltage is reduced by a factor of $1/\sqrt{2}$ from the magnitude of the

input source), and the asymptotic behavior for large frequencies (see Problem 9). The consideration of each of these special cases both adds to our understanding of this problem and confirms the utility of the general expressions obtained in Step 4.

Special Case 1: Direct Current ($\omega = 0$)

The frequency of a DC source is $\omega = 0$ so the impedance is zero for each inductor and is infinite for the capacitor. Thus, referring to Figure 6-2, the low-pass filter is seen to simply transmit the input unchanged through to the output port since the inductors become short-circuited wires and the capacitor becomes an open-circuit at DC. Thus, this filter should exhibit a gain of 1 and no phase shift.

Recall that the gain and phase shift are, in general, given by

> **M;**

$$\sqrt{\frac{R^2}{\omega^6 L^4 C^2 + (-4L^3 C + R^2 C^2 L^2)\omega^4 + (4L^2 - 2R^2 CL)\omega^2 + R^2}}$$

> **theta;**

$$\text{argument}\left(-\frac{jR}{-jR + jR\omega^2 CL + 2\omega L - \omega^3 L^2 C}\right)$$

In this case it is clear that the gain is $M=1$ and the phase shift is $\theta = \arctan(0) = 0$. The same results are obtained by Maple:

> **gain = simplify(subs(omega=0, M));**

$$gain = 1$$

> **phaseshift = simplify(subs(omega=0, theta));**

$$phaseshift = 0$$

Special Case 2: Natural Frequency ($\omega = \omega_0$)

The natural frequency of the circuit occurs when $\omega = \omega_0$. The gain and phase shift in this case are

> **gain = simplify(subs(omega=omega0, M));**

$$gain = R \sim \sqrt{\frac{C}{L}}$$

> **phaseshift = simplify(subs(omega=omega0, theta));**

$$phaseshift = -\frac{1}{2}\pi$$

Observe that though the gain of a low-pass filter at the natural frequency depends on the specific values of the circuit parameters, the phase shift is always -90° .

Special Case 3: 3-dB Frequencies

A 3-dB point (or frequency) of the circuit is any frequency at which the magnitude of the output voltage is reduced by a factor of $\sqrt{2}$ from the magnitude of the input voltage, that is, frequencies where

$M(\omega) = \frac{M(0)}{\sqrt{2}} = \frac{1}{\sqrt{2}}$. Note that $20 \log_{10}\left(\frac{1}{\sqrt{2}}\right) = -3.01$ dB, which explains why the 3 dB points represent those frequencies where the voltage amplitude is decreased by a factor of $\frac{1}{\sqrt{2}}$.

Since power is proportional to voltage squared, the 3-dB points for an LP filter also represent those frequencies where the transmitted power is reduced by a factor of 2 from DC.

In light of the square root in the gain, it is advisable to work with the square of this equation:

$$> \text{EQ3dB} := M^2 = 1/2;$$

$$\text{EQ3dB} := \frac{R^2}{\omega^6 L^4 C^2 + (-4L^3 C + R^2 C^2 L^2)\omega^4 + (4L^2 - 2R^2 C L)\omega^2 + R^2} = \frac{1}{2}$$

You could simply ask Maple to solve this equation for ω , but this is likely to return a total of six solutions. (Do you see why?) The number of solutions can be cut in half simply by noting that the denominator is an even function of ω . Let $\mu = \omega^2$ and then solve for μ . Each real and positive root, μ , will lead to a 3-dB frequency, $\omega = \sqrt{\mu}$.

The change of variables leads to the equation

$$> \text{EQ3dB2} := \text{subs}(\omega = \text{sqrt}(\mu), \text{EQ3dB});$$

$$\text{EQ3dB2} := \frac{R^2}{\mu^3 L^4 C^2 + (-4L^3 C + R^2 C^2 L^2)\mu^2 + (4L^2 - 2R^2 C L)\mu + R^2} = \frac{1}{2}$$

The solutions to this equation are



```
> MU := [ solve( EQ3dB2, mu ) ];
```

$$MU := \left[\begin{array}{l} \frac{\frac{1}{6} \%1^{1/3} + \frac{2}{3} \%2 - \frac{1}{3} \frac{-4L \sim + R \sim^2 C \sim}{C \sim L \sim}}{L \sim}, \\ -\frac{\frac{1}{12} \%1^{1/3} - \frac{1}{3} \%2 - \frac{1}{3} \frac{-4L \sim + R \sim^2 C \sim}{C \sim L \sim} + \frac{1}{2} j\sqrt{3} \left(\frac{1}{6} \%1^{1/3} - \frac{2}{3} \%2 \right)}{L \sim}, \\ -\frac{\frac{1}{12} \%1^{1/3} - \frac{1}{3} \%2 - \frac{1}{3} \frac{-4L \sim + R \sim^2 C \sim}{C \sim L \sim} - \frac{1}{2} j\sqrt{3} \left(\frac{1}{6} \%1^{1/3} - \frac{2}{3} \%2 \right)}{L \sim} \end{array} \right]$$

$$\%1 := -64L \sim^3 + 156R \sim^2 C \sim L \sim^2 + 24L \sim R \sim^4 C \sim^2 - 8R \sim^6 C \sim^3$$

$$+ 12R \sim \sqrt{-3 \frac{32L \sim^3 - 35R \sim^2 C \sim L \sim^2 - 28L \sim R \sim^4 C \sim^2 + 8R \sim^6 C \sim^3}{C \sim}} C \sim L \sim$$

$$\%2 := \frac{4L \sim^2 - 2R \sim^2 C \sim L \sim + R \sim^4 C \sim^2}{C \sim L \sim \%1^{1/3}}$$

At first glance it appears that this equation has one real root and two complex-valued roots. However, the expressions are sufficiently complicated that it is difficult to be certain about this. One test is to evaluate the three roots using $L = 10$ H, $C = 6$ μ F, and $R = 1$ k Ω :

```
> MU2 := sort( evalf( subs( PARAM, MU ) ) );
```

```
MU2 := [5701.533437, 12748.24941, 38216.88382]
```

Since there are, in fact, three positive values of μ that satisfy this equation, there will be three separate 3-dB frequencies:

```
> FREQ3dB := map( sqrt, MU2 );
```

```
FREQ3dB := [75.50849910, 112.9081459, 195.4913907]
```

These results are consistent with the corrected Bode amplitude plot: there are three points on this curve where the gain is -3 dB. Further verification of these results requires conversion from frequency into the normalized variable $\log_{10}(\omega/\omega_0)$:

```
> CONVERT := omega -> evalf( subs( PARAM, log[10](omega/omega0) ) );
```


```
CONVERT := omega -> evalf( subs( PARAM, log10( (omega/omega0) ) ) )
```

```
> map( CONVERT, FREQ3dB );
```

```
[-.2329285368, -.05819909883, .1802032616]
```

A graphical verification of the 3-dB points is also possible:

```
> plots[display]( [ BODEp , plot(-3, color=GREEN) ],
>                 view=[-0.3..0.3, -10..5], xtickmarks=5,
>                 title='3dB frequencies' );
```



This example should not be used to conclude that all LP filters have more than one 3-dB point. Though a full analysis of 3-dB points is beyond the scope of this discussion, note that the cutoff frequency is very close to the mean of ω_0 and the middle 3-dB point.

What If



Suppose you need to design a low-pass filter whose cut-off frequency, ω_c , is increased by a factor of 10. How would the value of L need to be changed if the original values for R and C are used? How would C be chosen if the original L and R values are used? How would R be chosen if the original L and C are used? Find all physical solutions and verify each result with an appropriate Bode plot.

Hint: The general equation that must be satisfied is

$$\frac{R^2}{\omega_{10}^6 L^4 C^2 + (-4L^3 C + R^2 C^2 L^2) \omega_{10}^4 + (4L^2 - 2R^2 CL) \omega_{10}^2 + R^2} = \frac{1}{2}$$

for an appropriate value of ω_{10} .

6-2 CALCULUS

Limits are the fundamental concept behind the two principal components of calculus: the derivative and the integral. The corresponding Maple commands are **limit**, **diff**, and **int** and their inert forms **Limit**, **Diff**, and **Int**. This brief introduction discusses only these fundamental concepts; the **student** package contains a large collection of additional calculus-based tools, including integration by parts, line integrals, and triple integrals.

The limit of an expression $f(x)$ as x approaches a , $\lim_{x \rightarrow a} f(x)$, is translated to Maple as **limit(f(x), x=a)**. An optional third argument can be used to specify one-sided limits.

EXAMPLE 6-4

Limits

Use Maple to evaluate each of the following limits:

$$(a) \quad \lim_{x \rightarrow -2} \frac{x^3 + 8}{x + 2}$$

$$(b) \quad \lim_{\theta \rightarrow \pi} \tan\left(\frac{\theta}{2}\right)$$

$$(c) \quad \lim_{\theta \rightarrow \pi^-} \tan\left(\frac{\theta}{2}\right)$$

$$(d) \quad \lim_{x \rightarrow 3} \frac{\sin(3x^2 - 27)}{-2x^3 + x^2 + 45}$$

$$(e) \quad \lim_{x \rightarrow 0} \sin\left(\frac{1}{x}\right)$$

$$(f) \quad \lim_{y \rightarrow x} \frac{\sin(y) - \sin(x)}{y - x}$$

SOLUTION

Each limit can be directly translated into Maple. The only special consideration is in (c), where the one-sided limit from the left is specified by including **left** as the third argument of the **limit** command

> **restart**;

(a) > **limit((x^3+8)/(x+2), x=-2);**

12

(b) > **limit(tan(theta/2), theta=Pi);**

undefined

(c) > **limit(tan(theta/2), theta=Pi, left);**

∞

(d) `> limit(sin(3*x^2-27)/(x^2-2*x^3+45), x=3);`

$$\frac{-3}{8}$$

(e) `> limit(sin(1/x), x=0);`

-1 .. 1

(f) `> limit((sin(y)-sin(x))/(y-x), y=x);`

cos(x)

The results from parts (b) and (c) illustrate the way Maple responds to limits that do not exist or are unbounded. The result in (e) should be interpreted to mean that the limit is not defined because the value of $\sin\left(\frac{1}{x}\right)$ oscillates within the interval $[-1,1]$ infinitely often near $x = 0$. The limit in (f) is recognized as the definition of the derivative of $\sin(x)$; in light of this observation, the result is exactly what you should expect.

.....

The **Limit** command is the *inert*, or unevaluated, form of **limit**. Except that the inert form of a function is not evaluated, it is identical to the regular form of the command. Two common usages of the inert form of a command are to verify that the correct quantity has been entered and to have more control over the evaluation of the expression. The **value** command forces evaluation of the inert form with the corresponding normal version of the command (for example, replace **Limit** with **limit** and then call **value**); the **evalf** command can be used if a purely numeric evaluation is desired.

EXAMPLE 6-5

Euler's constant, e

Euler's constant, e, can be defined as $\lim_{n \rightarrow \infty} \left(1 + \frac{1}{n}\right)^n$. Use the inert form of **limit** to evaluate this limit both symbolically and numerically using 20 significant digits.

SOLUTION

The definition of e, as an unevaluated limit, is

`> eLIM := Limit((1+1/n)^n, n=infinity);`

$$eLIM := \lim_{n \rightarrow \infty} \left(1 + \frac{1}{n}\right)^n$$

The normal evaluation of this inert expression is forced using the **value** command.

`> eLIM = value(eLIM);`

$$\lim_{n \rightarrow \infty} \left(1 + \frac{1}{n}\right)^n = e$$

The floating-point evaluation of the same limit, using 20 digits in the computations, can be obtained using **evalf**:

```
> eLIM = evalf( eLIM, 20 );
```

$$\lim_{n \rightarrow \infty} \left(1 + \frac{1}{n} \right)^n = 2.7182818284590452354$$

.....



Try It

Note how the presentation of the limit and its value in Example 6-5 is more informative than the format used in Example 6-4. Redo each of the problems in Example 6-4 using a style similar to that introduced in Example 6-5.

EXAMPLE 6-6

Asymptotes of a Rational Function

Find all asymptotes (vertical, horizontal, and oblique) for the function

$f(x) = \frac{2x^2 - 7x + 1}{3 - x}$. Create a plot that clearly shows all interesting features of f .

SOLUTION

The function could be entered as a Maple expression or, using the arrow operator, as a function. For a change of pace, and in preparation of later examples in this chapter, the arrow operator will be used here:

```
> f := x -> (2*x^2-7*x+1)/(3-x);
```

$$f := x \rightarrow \frac{2x^2 - 7x + 1}{3 - x}$$

The singularity in the denominator when $x = 3$ suggests the probable existence of a vertical asymptote at $x = 3$. To confirm this, it is necessary to check that at least one of the left and right one-sided limits is unbounded at $x = 3$:

```
> limR := Limit( f(x), x=3, right );
```

```
> limR = value( limR );
```

$$\lim_{x \rightarrow 3^+} \frac{2x^2 - 7x + 1}{3 - x} = \infty$$

```
> limL := Limit( f(x), x=3, left );
```

```
> limL = value( limL );
```

$$\lim_{x \rightarrow 3^-} \frac{2x^2 - 7x + 1}{3 - x} = -\infty$$

Thus, $x = 3$ is a vertical asymptote for f .

Since the degree of the numerator exceeds the degree of the denominator, there are no horizontal asymptotes to this function. But there should be an oblique asymptote. Long division is the usual technique used to manually compute an oblique asymptote. The Maple commands **quo** and **rem** can be used to obtain this information. The oblique asymptote can be obtained from the quotient of the numerator and denominator of f as follows:

```
> NN := numer( f(x) );
> DD := denom( f(x) );
```

$$\begin{aligned} NN &:= -2x^2 + 7x - 1 \\ DD &:= -3 + x \end{aligned}$$

```
> obliq_asymp := quo( NN, DD, x, 'rest' );
```

$$\text{obliq_asymp} := -2x + 1$$

Observe how the optional fourth argument to **quo** is used to obtain both the remainder and quotient simultaneously. Thus, an equivalent representation of the function should be

```
> fEQUIV := obliq_asymp + rest/DD;
```

$$fEQUIV := -2x + 1 + \frac{2}{-3 + x}$$

One way to check this result is to normalize the difference between the two representations of the function:

```
> simplify( fEQUIV - f(x) );
```

$$0$$

Since this difference is zero, the two representations are equivalent.

All that remains is to verify that $y = -2x + 1$ is an oblique asymptote for f :

```
> obliqLIM1 := Limit( f(x)-obliq_asymp, x=infinity );
```

```
> obliqLIM1 = value( obliqLIM1 );
```

$$\lim_{x \rightarrow \infty} \frac{2x^2 - 7x + 1}{3 - x} + 2x - 1 = 0$$

```
> obliqLIM2 := Limit( f(x)-obliq_asymp, x=-infinity );
```

```
> obliqLIM2 = value( obliqLIM1 );
```


$$\lim_{x \rightarrow (-\infty)} \frac{2x^2 - 7x + 1}{3 - x} + 2x - 1 = 0$$

Of course, these results can be seen by inspection from the fact that $f(x) - (-2x + 1) = \frac{2}{x - 3}$. (Three alternative methods for finding the oblique asymptote are presented in Problem 3 in this chapter.)

The last step is to create a plot of the function f . Since f is discontinuous, you know that it is necessary to use the **discont=true** option and to provide an appropriate vertical range:

```
> plot( f(x), x=-10..10, -50..50, discontin=true,
>       title=`f(x) = ( 2*x^2 - 7*x + 1 )/( 3 - x )` );
```

$$f(x) = (2*x^2 - 7*x + 1)/(3 - x)$$



.....

The derivative of an expression, **EXPR**, with respect to the name **X**, is obtained with the command **diff(EXPR, X)**;. The inert form of **diff** is **Diff**. A third differentiation command, which is used to compute derivatives of functions (not expressions), is **D**.

EXAMPLE 6-7

Derivatives

Compute the derivative of $f(x) = \sqrt{x} + \sin(2x) + e^{\frac{1}{x}} + \frac{x^2 + 3x}{x + 1}$ using both **diff** and **D**. Use **Diff** to improve the appearance of the results.

SOLUTION

Since both **diff** and **D** are to be used, it will be necessary to implement the function using the arrow operator:

```
> f := x -> sqrt(x) + sin(2*x) + exp(1/x) + (x^2+3*x)/(x+1);
```

$$f := x \rightarrow \sqrt{x} + \sin(2x) + e^{\left(\frac{1}{x}\right)} + \frac{x^2 + 3x}{x + 1}$$

The unevaluated derivative is obtained using **Diff** with the expression **f(x)**; evaluation is forced using the **value** command.

> **df := Diff(f(x), x);**

$$df := \frac{\partial}{\partial x} \left(\sqrt{x} + \sin(2x) + e^{\left(\frac{1}{x}\right)} + \frac{x^2 + 3x}{x+1} \right)$$

> **df = value(df);**

$$\frac{\partial}{\partial x} \left(\sqrt{x} + \sin(2x) + e^{\left(\frac{1}{x}\right)} + \frac{x^2 + 3x}{x+1} \right) = \frac{1}{2} \frac{1}{\sqrt{x}} + 2\cos(2x) - \frac{e^{\left(\frac{1}{x}\right)}}{x^2} + \frac{2x+3}{x+1} - \frac{x^2+3x}{(x+1)^2}$$

To produce the same result using **D**, it is necessary to first compute the function (not expression) that is the derivative of the function **f**.

> **Df := D(f);**

$$Df := x \rightarrow \frac{1}{2} \frac{1}{\sqrt{x}} + 2\cos(2x) - \frac{e^{\left(\frac{1}{x}\right)}}{x^2} + \frac{2x+3}{x+1} - \frac{x^2+3x}{(x+1)^2}$$

Since **Df** is a function, the value of this function is obtained by supplying an argument. Thus, another way to present the derivative is

> **df = Df(x);**

$$\frac{\partial}{\partial x} \left(\sqrt{x} + \sin(2x) + e^{\left(\frac{1}{x}\right)} + \frac{x^2 + 3x}{x+1} \right) = \frac{1}{2} \frac{1}{\sqrt{x}} + 2\cos(2x) - \frac{e^{\left(\frac{1}{x}\right)}}{x^2} + \frac{2x+3}{x+1} - \frac{x^2+3x}{(x+1)^2}$$

.....

Note that Maple displays all derivatives using the standard notation for partial derivatives. This is something that you need to accept; it cannot be changed.

Second- and higher-order derivatives can be found by specifying additional arguments to **diff** or **Diff**. For example, the second and fifth derivatives of **EXPR** with respect to **X** can be obtained with **diff(EXPR, X, X)** and **diff(EXPR, X \$ 5)**, respectively. The corresponding second and fifth derivative functions obtained with **D** would be **D(D(f))** and, using the repeated composition operator **@**, **(D @ 5)(f)**.



Compute the first three derivatives of the function $f(x) = e^{-1/x^2}$ using both **diff** and **D**. (See also Problem 4 in this chapter).

Note the close correspondence between Maple syntax and mathematical notation and terminology. This connection is particularly strong for the

integration commands, **int** and **Int**. Example 6-8 provides illustrations of indefinite integrals. Proper and improper definite integrals are examined in Example 6-9. Integrals of piecewise-defined functions are discussed in Examples 6-10 and 6-11. The Strength and Toughness application from Chapter 2 is revisited in Examples 6-10 and 6-11. Example 6-12, involving numerical integration, illustrates the increased efficiency that can result from the intelligent use of the inert form of a command.

EXAMPLE 6-8**A Trigonometric Integral**

Evaluate the indefinite integral $\int \sin(x)\cos(bx)dx$ for all real numbers b .

SOLUTION

The inert integration command is used here to, once again, improve the clarity of the results.

```
> indefINT := Int( sin(x)*cos(b*x), x );
> indefINT = value( indefINT );
```

$$\int \sin(x)\cos(bx)dx = -\frac{1}{2} \frac{\cos((1+b)x)}{1+b} + \frac{1}{2} \frac{\cos((-1+b)x)}{-1+b}$$

Observe that Maple does not include the constant of integration. Thus, it is your responsibility to include an appropriate constant of integration, using a name selected by the user, whenever it is needed. This will be illustrated in all subsequent examples.

Note that this result cannot be applied when $b = 1$ or $b = -1$. Since $\cos(-x) = \cos(x)$, cosine is an even function, and the results for the two special cases should be the same. The indefinite integrals when $b = 1$ and $b = -1$ are found to be

```
> indefINT1 := subs( b=1, indefINT );
> indefINT1 = value( indefINT1 ) + C;
```

$$\int \sin(x)\cos(x)dx = \frac{1}{2} \sin(x)^2 + C$$

```
> indefINT2 := subs( b=-1, indefINT );
> indefINT2 = value( indefINT2 ) + C;
```

$$\int \sin(x)\cos(x)dx = \frac{1}{2} \sin(x)^2 + C$$

Try It

The indefinite integrals when $|b| = 1$ are easily found using the substitution $u = \sin(x)$. The integrals could also be found by using the substitution

$v = \cos(x)$ or by noting that $\sin(x)\cos(x) = \frac{\sin(2x)}{2}$. Compute the indefinite

integral $\int \frac{\sin(2x)}{2} dx$. Does this yield the same result as in Example 6-8?

Are the results equivalent? Explain.

The only difference in Maple between an indefinite and definite integral is the specification of the interval of integration. For example, the integral of a function $f(x)$ over the interval $[a, b]$ would be entered as `int(f(x), x = a .. b)`.

EXAMPLE 6-9**Definite Integrals**

Evaluate the definite integrals $\int_{-2}^2 f(x)dx$ and $\int_0^3 f(x)dx$, where

$$f(x) = \frac{2x^2 - 7x + 1}{3 - x}. \text{ Note that the second integral is an improper integral.}$$

SOLUTION

The function f is given by

```
> f := x -> (2*x^2-7*x+1)/(3-x);
```

$$f := x \rightarrow \frac{2x^2 - 7x + 1}{3 - x}$$

The definite integral over the interval $[-2, 2]$ can be evaluated as follows:

```
> fINT1 := Int( f(x), x=-2 .. 2 );
```

```
> fINT1 = value( fINT1 );
```

$$\int_{-2}^2 \frac{2x^2 - 7x + 1}{3 - x} dx = 4 - 2\ln(5)$$

Since f has a singularity at $x = 3$, the second integral is improper. Thus, the integral converges if and only if $\lim_{\alpha \rightarrow 0^+} \int_0^{3-\alpha} f(x)dx$ converges. Maple automatically detects the singularity and evaluates the appropriate limit, when possible, for improper integrals. Thus, as far as Maple is concerned, the user does not have to change the way in which the integral is computed:

```
> fINT2 := Int( f(x), x=0 .. 3 );
```

```
> fINT2 = value( fINT2 );
```

$$\int_0^3 \frac{2x^2 - 7x + 1}{3 - x} dx = -\infty$$

Thus, this integral diverges to $-\infty$.

To better understand the steps that are involved in arriving at the value of an improper integral, let $\alpha > 0$ and consider the proper integral

```
> fINT2a := Int( f(x), x=0 .. 3-alpha );
```

```
> fINT2a = value( fINT2a );
```

$$\int_0^{3-\alpha} \frac{2x^2 - 7x + 1}{3 - x} dx = -6 + 5\alpha - \alpha^2 + 2\ln(-\alpha) - 2\ln(3) - 2j\pi$$

This is a very strange result. How did a nonzero imaginary term arise in this problem? When limits are taken using this result, α will be positive; what is $\ln(-\alpha)$? The answers to both questions are related, and will be discussed further in the next Try It! exercise. Nonetheless, as α decreases to zero, the value of these definite integrals diverges to $-\infty$:

```
> impINT := Limit( fINT2a, alpha=0, right );
> fINT2 = impINT;
> `` = value( impINT );
```

$$\int_0^3 \frac{2x^2 - 7x + 1}{3 - x} dx = \lim_{\alpha \rightarrow 0^+} \int_0^{3-\alpha} \frac{2x^2 - 7x + 1}{3 - x} dx = -\infty$$

**Try It**

Repeat the evaluation of the proper integrals on the interval $[0, 3 - \alpha]$ after telling Maple that α is assumed to be positive. Does this change the value of these integrals? Does it change the convergence of the improper integral? What does this suggest about the value of $\ln(-\alpha)$?

EXAMPLE 6-10**Toughness as Three Integrals**

In Application 2 in Chapter 2, we saw that the toughness of a material is defined as being the area under the stress–strain curve. In this application, the area was found either geometrically or by an approximate method (via the **rightbox** command). Since the area under a curve can be expressed as a definite integral, the toughness can be found by integrating the stress over all strains up to the fracture point. For the specific material analyzed in Chapter 2, this means $toughness = \int_0^{0.074} \sigma(\epsilon) d\epsilon$ where the stress is 12000ϵ for $0 \leq \epsilon \leq 2.8\%$, 330 for $2.8\% \leq \epsilon \leq 3.2\%$, and $-210000\epsilon^2 + 23000\epsilon - 210$ for $3.2\% \leq \epsilon \leq 7.4\%$. Compute the toughness of the material analyzed in Chapter 2.

SOLUTION

The three parts of the stress function are

```
> stress[1] := 12000*epsilon;
```

$$stress_1 := 12000\epsilon$$

```
> stress[2] := 330;
```

$$stress_2 := 330$$

```
> stress[3] := -210000*epsilon^2 + 23000*epsilon - 210;
```

$$stress_3 := -210000\epsilon^2 + 23000\epsilon - 210$$

The toughness of this material is found as the sum of three definite integrals.

```

> toughness := Int( stress[1], epsilon=0..0.028 )
>           + Int( stress[2], epsilon=0.028..0.032 )
>           + Int( stress[3], epsilon=0.032..0.074 ):
> toughness = value( toughness );

```

$$\int_0^{.028} 12000\epsilon \, d\epsilon + \int_{.028}^{.032} 330 \, d\epsilon + \int_{.032}^{.074} -210000\epsilon^2 + 23000\epsilon - 210 \, d\epsilon = 22.33008000$$

To two significant digits, the toughness of this material is 22. Observe that this is well within the range found at the end of Step 3 of the analysis in Chapter 2.

.....

The specification of the stress–strain curve in three separate parts is a natural part of this problem. This did not impede the determination of the toughness, but it does complicate the generation of a plot of the stress–strain curve.

The **piecewise** command can be used to specify, in a single Maple object, all the different pieces of the function. The arguments to **piecewise** come in pairs of expressions where the first expression in each pair is Boolean (evaluates to either **true** or **false**). For example, the piecewise definition of the stress–strain curve is

```

> stressPW := piecewise( epsilon<0.028, stress[1],
>                       epsilon<0.032, stress[2],
>                       epsilon<0.074, stress[3] );

```

$$stressPW := \begin{cases} 12000\epsilon & \epsilon < .028 \\ 330 & \epsilon < .032 \\ -210000\epsilon^2 + 23000\epsilon - 210 & \epsilon < .074 \end{cases}$$

Note the convenient printed format—exactly what would be found in any standard mathematical discussion.

If ϵ has a value, the value returned by **piecewise** is the expression corresponding to the first Boolean expression that evaluates to **true**. (See the online help worksheet for **piecewise** for an explanation of the return value when none of the Boolean expressions are true.) Do not despair. This sounds more confusing than it is. To be sure you understand how this works, consider the following examples:

```

> eval( subs( epsilon = 0, stressPW ) );

```

0

```

> eval( subs( epsilon = 0.01, stressPW ) );

```

120.00

```

> eval( subs( epsilon = 0.032, stressPW ) );

```

310.960000

```
> eval( subs( epsilon = 0.10, stressPW ) );
0
```

The last result is zero because this is the default value returned from a piecewise-defined function when none of the Boolean expressions evaluates to **true**.



Use the **piecewise** definition of stress and strain to create a plot of the stress-strain curve for the material analyzed in Chapter 2.

EXAMPLE 6-11

Toughness as a Single Integral

Find the toughness of the material by the integration of the piecewise-defined stress function.

SOLUTION

```
> toughnessPW := Int( stressPW, epsilon=0..0.074 );
> toughnessPW = value( toughnessPW );
```

$$\int_0^{.074} \begin{cases} 12000 \epsilon & \epsilon < .028 \\ 330 & \epsilon < .032 \\ -210000\epsilon^2 + 23000\epsilon - 210 & \epsilon < .074 \\ 0 & otherwise \end{cases} d\epsilon = 22.33008000$$

Note that this value is identical to the one obtained in Example 6-10.

.....



The piecewise-defined stress function used in Example 6-11 is not quite correct. Note that although the stress is forced to be zero for strains after failure, no similar efforts have been made to prevent erroneous values for negative strains. Redefine the stress function so that this potential problem is avoided. Further, for the sake of simplifying evaluation, use the **unapply** command to make the stress a function of the strain. (See Chapter 3, or a help worksheet, for information about the **unapply** command) Plot this function over the domain $-0.01 < \epsilon < 0.10$.

To conclude this introduction to the use of Maple for calculus-based manipulations, consider the evaluation of the definite integral $\int_0^1 e^{-x^3} dx$. This is an example of an integral that exists (it is the integral of a continuous and bounded function on a finite interval) but cannot be written in terms of elementary functions.



The first objective is to understand that the mathematical theory guarantees that the existence of this integral is independent of our ability to express the value in terms of elementary functions. First, plot $y = e^{-x^3}$ on an appropriate interval and identify $\int_0^1 e^{-x^3} dx$ as the area of a plane region.

Then, use the plot to estimate the value of $\int_0^1 e^{-x^3} dx$.

Once Maple determines that it is unable to evaluate an expression, the expression is generally returned unevaluated. A floating-point approximation can be forced with the use of **evalf**.

EXAMPLE 6-12

Numerical Integration

- Verify that Maple is unable to evaluate the definite integral $\int_0^1 e^{-x^3} dx$.
- Use the **time** command to compare the CPU time required to obtain a numerical approximation to the integral using the two commands **evalf(int(exp(-x^3), x=0..1))**; and **evalf(Int(exp(-x^3), x=0..1))**;

SOLUTION

- Maple's inability to evaluate the integral will be detected when **value** attempts to evaluate the inert form of the integral.

```
> defINT := Int( exp( -x^3 ), x = 0 .. 1 );
> defINT = value( defINT );
```

$$\int_0^1 e^{-x^3} dx = \int_0^1 e^{(-x^3)} dx$$

- The following segments of Maple code can be used to test the execution time for the two different expressions given in the problem statement. Note that the **restart** commands are used to prevent any residual information from the first evaluation from influencing the execution speed of the second form.

```
> restart;
> tstart := time():
> area := evalf( int( exp( -x^3 ), x = 0 .. 1 ) ):
> tend := time():
> printf( '\n elapsed time: %6.3f seconds (area = %12.10f)\n ',
>         tend-tstart, area );
```

elapsed time: 3.174 seconds (area = .8075111821)

```

> restart;
> tstart := time():
> area := evalf( Int( exp( -x^3 ), x = 0 .. 1 ) ):
> tend := time():
> printf( '\n elapsed time: %6.3f seconds (area = %12.10f)\n ',
>         tend-tstart, area );

```

elapsed time: .731 seconds (area = .8075111821)

Note that the values of the integrals are the same up to (at least) 10 decimal digits, but that the form that used **Int** was more than four times as fast as the evaluation that used **int**. The difference in execution time is a result of the fact that Maple attempted to evaluate the **int** command before it realized that a numerical integration was necessary.

The appearance of the output has been improved with the use of the **printf** command (see the online help for **printf**).

.....

You should note that if the integral can be evaluated, then **evalf(int(...))** returns a floating-point approximation to the evaluated expression; no numerical integration is performed. To ensure the numerical evaluation of an integral, **evalf(Int(...))** should be used.

6-3 DIFFERENTIAL EQUATIONS

The applications in Chapters 3 and 5 have used differential equations to model different aspects of an engineering problem. As these applications illustrate, an ability to work with differential equations is essential for many engineers. For these reasons, differential equations are the final mathematical topic to be discussed in this chapter. Since knowledge of differential equations is not typically a prerequisite for courses that would use this module, this introduction will be brief. You are encouraged to revisit this section of the text when you take a course in differential equations.

A *differential equation* is an equation that includes the derivative of one or more unknown functions. The typical objective is to solve the differential equation—that is, to find the unknown functions that satisfy the differential equation (and any associated initial or boundary conditions). In many cases, it is not possible to find the solution to a differential equation. In these instances, the objective may be to use the differential equation to provide a qualitative description of the solution.

In terms of Maple, the derivatives in a differential equation are entered using **diff**, **Diff**, or **D**; the **dsolve** command is used to find exact (symbolic) and approximate (numeric) solutions to a differential equation; and the **DEtools** package, in particular the **DEplot** command, provides a variety of options for obtaining graphical information about a solution to a differential equation.

The water quality application in Chapter 5 is based on two differential equations. The BOD and Streeter-Phelps equations are

$$\begin{aligned}L'(t) &= -k_d L(t) \\ D'(t) &= k_d L(t) - k_r D(t)\end{aligned}$$

You are also provided with the solution to this system of equations, with initial conditions $L(0) = BOD_u$ and $D(0) = D_0$:

$$\begin{aligned}L(t) &= BOD_u e^{-k_d t} \\ D(t) &= \frac{k_d}{k_r - k_d} BOD_u (e^{-k_d t} - e^{-k_r t}) + D_0 e^{-k_r t}\end{aligned}$$

Recall that **D** is Maple's differentiation operator, so it cannot be used as the name for the oxygen deficit; a simple modification that was used in Chapter 5 is **DD**.

In Examples 6-13 and 6-14 you will see how **dsolve** can be used to obtain the general and particular solution to a differential equation. Additional examples and information about the use of Maple to analyze differential equations can be found in the Maple's online help.

EXAMPLE 6-13

General Solution to a Differential Equation

Use **dsolve** to find the general solution to the BOD equation; then find the solution that satisfies the initial condition $L(0) = BOD_u$.

SOLUTION

The BOD differential equation is entered as

```
> BODeqn := diff( L(t), t ) = -kd * L(t);
```

$$BODeqn := \frac{\partial}{\partial t} L(t) = -kdL(t)$$

The syntax for **dsolve** is very similar to the syntax for **solve** and **fsolve**. The general solution of an equation can be found by specifying the differential equation and unknown function as arguments to **dsolve**:

```
> BODgen := dsolve( BODeqn, L(t) );
```

$$BODgen := L(t) = e^{(-kdt)} _C1$$

It is not difficult to visually confirm that these functions are solutions to the BOD differential equation for any value of the constant **_C1**. In more complicated equations it might be more convenient to use Maple to perform this check. One way to verify that a function satisfies a differential equation is to substitute the solution into the differential equation with all terms moved to one side of the equation:

```
> CHECK := subs( BODgen, lhs(BODeqn) - rhs(BODeqn) );
```

$$CHECK := \left(\frac{\partial}{\partial t} e^{(-kdt)} _C1 \right) + kd e^{(-kdt)} _C1$$

then force full evaluation of the resulting expression

```
> eval( CHECK );
```

0

Since the result is zero, or an expression that reduces to zero, the equation is satisfied and the function is a solution.

The particular function that also satisfies the initial condition, $L(0) = BOD_u$, can be found by substituting the initial condition (and initial time) into the equation for the general solution, then solving for the constant(s). For example, for the BOD problem:

```
> BODic := L(0) = BODu;
```

$$BODic := L(0) = BODu$$

```
> BODinit := subs( t=0, BODic, BODgen );
```

$$BODinit := BODu = e^0 _C1$$

```
> solve( BODinit, { _C1 } );
```

$$\{ _C1 = BODu \}$$

Thus, the initial condition is satisfied if and only if $_C1 = BOD_u$, and the only function that satisfies the BOD differential equation and the initial condition is

```
> BODsoln := subs( ", BODgen );
```

$$BODsoln := L(t) = e^{(-kdt)} BODu$$

.....

A simpler means of obtaining the solution to an initial value problem is to call **dsolve** with a first argument that is a set containing both the differential equation and the initial condition.

EXAMPLE 6-14

Particular Solution to a Differential Equation

Use the method just described to find the particular solution to the BOD equation.

SOLUTION

Using the assignments made in Example 6-13, the set containing the differential equation and initial condition is

> **BODivp** := { **BODeqn**, **BODic** };

$$BODivp := \left\{ \frac{\partial}{\partial t} L(t) = -kd L(t), L(0) = BODu \right\}$$

Then, the particular solution can be found with the single command

> **dsolve**(**BODivp**, **L(t)**);

$$L(t) = e^{(-kdt)} BODu$$

This solution is, by inspection, the same as the one found in Example 6-13.

.....



Try It Find the general and particular solutions to the Streeter-Phelps equation, with initial condition $D(0) = D_0$. *Hint:* Begin by substituting the solution to the BOD equation into the Streeter-Phelps equation:

> **DOcons** := **DO(t)** + **DD(t)** = **DOsat**;

$$DOcons := DO(t) + DD(t) = DOsat$$

> **DOeqn** := **diff**(**isolate**(**DOcons**, **DO(t)**), **t**);

$$DOeqn := \frac{\partial}{\partial t} DO(t) = -\left(\frac{\partial}{\partial t} DD(t) \right)$$

> **DOeqn** := **subs**(**SPEqn**, **DOeqn**);

$$DOeqn := \frac{\partial}{\partial t} DO(t) = -kdL(t) + kr DD(t)$$

> **DOeqn** := **subs**(**isolate**(**DOcons**, **DD(t)**), **DOeqn**);

$$DOeqn := \frac{\partial}{\partial t} DO(t) = -kdL(t) + kr(-DO(t) + DOsat)$$

SUMMARY

This chapter demonstrated how Maple can be used in problems that involve complex numbers, calculus, and differential equations. The application used nodal analysis to investigate a low-pass electrical filter, including the determination of the circuit's frequency response. Expressions for the voltage transfer ratio amplitude and phase were calculated and plotted. The calculus section focused on problems that involve limits, derivatives, and integrals. Several ways that inert commands can be used to improve both appearance and efficiency were introduced and demonstrated. Some of the examples revisited applications introduced in earlier chapters. In particular, the toughness of a material was computed by direct integration after the stress-strain curve was defined as a piecewise-defined function and the biochemical oxygen demand (BOD) and oxygen deficit levels in the water quality application were obtained by solving the BOD and Streeter-Phelps equations.

Keywords

argument	initial condition
asymptotic expansion	initial value problem
calculus	limit
conjugate	magnitude (modulus)
definite integral	numerical integration (quadrature)
derivative	one-sided limit
differential equation	partial fraction decomposition
differentiation	particular solution.
general solution	piecewise-defined function
improper integral	qualitative analysis
indefinite integral	series expansion
inert (unevaluated) form	

Maple Commands

@@ (repeated composition operator)	int, Int
abs, conjugate, Re, Im	intparts (from the student package)
alias	isolate
allvalues, RootOf	limit, Limit
argument	map
asympt, series, convert, parfrac	piecewise
complexplot, complexplot3d (from plots package)	printf
DEplot (from the DEtools package)	quo, rem
diff, Diff, D	semilogplot (from the plots package)
dsolve	surd
evalc,	time
I	

References

1. Lopez, R., *Maple via Calculus: A Tutorial Approach*, New York: Birkhauser, 1994.
2. Crawford, R.S. Jr., *Waves*, New York: McGraw-Hill, 1968, pp. 128-130.
3. Johnson, D.E., Hilburn, J.L, Johnson, J.R., and Scott, P.D., *Basic Electric Circuit Analysis*, 5th ed. Englewood Cliffs, N.J.: Prentice-Hall, 1995, pp. 284-308, 322-327.

Problems

1. (a) Find the eighth roots of 10. Use **nops** to verify that the correct number of roots have been found.
 (b) Create separate lists containing the real-valued roots and the complex-valued roots. *Hint:* The **select** and **remove** commands are recommended for creating subsets of lists or sets as requested in this problem. Further, a common method for testing if an object is complex valued is to use the Boolean function **has** to check if the object contains **I**. For example, **has(f, I)**; returns **true** if **f** contains **I** and otherwise returns **false**.
2. The modulus and argument of a complex number can be used to specify a complex number in polar coordinates. This conversion can be accomplished using Maple's **polar** command, whereas conversion from polar to standard form is obtained using **evalc**. (For additional details, see the help worksheet for **polar**.)

Let a be a positive number. Find the polar representations of the square, cube, and fourth roots of a . Without using **solve**, what are the fifth roots of a ?

3. Let f be the function introduced in Example 6-6, that is,

$$f(x) = \frac{2x^2 - 7x + 1}{3 - x}.$$

- (a) Use **asympt** to find the asymptotic expansion of f .
- (b) Use **series** to find the series expansion of f about both $x = 0$ and $x = 3$.
- (c) Use **convert, parfrac** to find the partial fraction decomposition of f .
- (d) Compare the results found in parts (a), (b), and (c). How hard is it to identify the oblique asymptote in each case? What other information is contained in these results?

Consult the appropriate help worksheets for information about the syntax and interpretation of results from each of these commands.

4. Note that the function $f(x) = e^{-1/x^2}$ is not defined at $x = 0$. (See also the Try It! exercise immediately following Example 6-7.)
 (a) Plot f and its first three derivatives on the interval $[-1, 1]$.
 (b) Is it possible to define $f(0)$ so that f is continuous at $x = 0$?
 (c) Use part (b), and the definition of the derivative, to compute $f'(0)$. Are any derivatives of f continuous at $x = 0$?

5. (a) Compute the definite integral of $g(x) = \frac{2x^2 - 7x + 1}{\sqrt{3-x}}$ and $h(x) = \frac{2x^2 - 7x + 1}{(3-x)^2}$. Cross-check your results using a graph and interpreting the integral as a (signed) area.
- (b) For what values of p does $\int_0^3 \frac{2x^2 - 7x + 1}{(3-x)^p} dx$ converge?
6. Use the **intparts** command from the student package to evaluate—step-by-step—the indefinite integrals (a) $\int x^2 \ln(x) dx$ and (b) $\int e^{ax} \cos(bx) dx$. Does it matter which term in the integrand is chosen for “u”?
7. The integral $\int_0^1 e^{-x^3} dx$ was found to be approximately 0.80751118 in Example 6-12. Another approach to this problem is to expand the integrand in a power series (see Problem 3 in this chapter), then integrating the resulting polynomial.
- (a) How many terms in the power series expansion of e^{-x^3} , centered at $x = 0$, are needed to approximate the value of this integral to five (5) decimal places?
- (b) How many terms are needed when the power series is based at $x = 1$? at $x = 1/2$?
- Hints:* The Maple command for computing power series is **series** (see also Problem 3 in this chapter). Use **convert, polynom** to convert the result returned by **series** to a polynomial before integrating.
8. Determine which of the following expressions satisfy the differential equation $u''(t) + 2u'(t) + 2u(t) = 0$:
- $$v_1(t) = e^t \sin(2t), \quad v_2(t) = e^{-t} \sin(t), \quad v_3(t) = e^{-t}, \quad v_4(t) = \cos(t).$$
9. (a) Another interesting frequency in the analysis of the low-pass filter is $\omega' = R/L$. Find the gain and phase shift for this frequency.
- (b) Determine the behavior of both the gain and phase shift as ω increases without bound.
- (c) Show that $M_{dB} \sim -60 \log_{10}(\omega) + 20(\log_{10}(\omega') + 2 \log_{10}(\omega_0))$ for “large” frequencies. (In this context, high-frequency means $\omega/\omega_0 \gg 1$ and $\omega/\omega' \gg 1$.)
- (d) Use the result from part (c) to show that the cut-off frequency, ω_c , satisfies $\omega_c^3 = \omega' \omega_0^2$.
- (*Hint:* Use a combination of **limit, series**, and/or **asympt.**)
10. a) Produce corrected Bode amplitude plots, and compute the cut-off frequencies when the load resistance is $R = 10\Omega$, $R = 1k\Omega$, $R = 500k\Omega$, $R = 1M\Omega$, $R = 10M\Omega$, and $R = 100M\Omega$. How does changing the load resistance value affect the operation of the low-pass filter?

- b) The corrected Bode phase plot is a plot of θ (in degrees) vs. $\log_{10}(\omega/\omega_0)$. Create the corrected Bode phase plot for the original problem and for each of the resistances in part (a).
- c) Present the plots in parts (a) and (b) as an animated sequence of plots with increasing values of resistance.
11. Suppose two identical low-pass filters are “cascaded.” Determine the frequency response function, $\frac{V_o}{V_i}$, for this new composite component. Does this new component better approximate an “ideal” low-pass filter? (*Hint:* The output voltage from the first low-pass filter becomes the input voltage to the second low-pass filter.)
12. Note that the solution to the Streeter-Phelps equation is not defined when $kd = kr$. One possible way to define the solution when $kd = kr$ is with limits. What function is defined by $\lim_{kd \rightarrow kr} D(t)$? Verify that this function satisfies the Streeter-Phelps equation when $kd = kr$.
13. In Chapter 3, Problem 6, you were told that an airplane’s range (s), lift (L), drag (D), thrust-specific fuel consumption ($TSFC$), take-off mass (m_0), and landing mass (m) satisfy the relationship

$$s = \frac{V \ln\left(\frac{m}{m_0}\right)L}{D g TSFC}$$

This relationship is obtained from Breguet’s equation

$$m'(s) = -\frac{g D m(s)}{V L TSFC} \text{ with initial condition } m(0) = m_0.$$

- (a) Use Maple to find the particular solution to Breguet’s equation that satisfies the given initial condition.
- (b) Note that Breguet’s equation gives the landing mass in terms of the other quantities. Use Maple to rewrite the solution to Breguet’s equation as $s = F(m, m_0, L, D, V, TSFC, g)$. Do you obtain the result stated above (and in Chapter 3)? (Explain any differences.)
- (c) The specific form in which the relationship is given depends largely on the way in which it will be used. Give examples of situations in which each of the two forms of the relationship discussed in this problem might be used.

